

Web-based Supplementary Materials for *Analysis of Capture-Recapture Models with Individual Covariates Using Data Augmentation* by J.A. Royle

Web Appendix A: WinBUGS model specifications for *Microtus* examples.

This appendix contains the *WinBUGS* model specifications for both the *Microtus* data (Fig. 1) and the waterfowl survey data (Fig. 2). These model specifications were executed from **R** using the *R2WinBUGS* library. In some instances, *WinBUGS* does not appear to update latent continuous random effects in some classes of capture-recapture models. This can be diagnosed by inspecting the MCMC history of the variance component of the random effect, which will show little or no movement from its initial value and chains that do not mix. In such cases, modification of the prior by truncation, according to

```
wt[i] ~ dnorm(mu.wt,tau.wt)I(-6,6)
```

appears to resolve the problem.

```

model {

  # Prior distributions for model parameters
  psi ~ dunif(0,1)
  mu.wt ~ dnorm(0,.001)
  tau.wt ~ dgamma(.001,.001)
  sigma ~ sqrt(1/tau.wt)
  a0 ~ dnorm(0,.001)
  a1 ~ dnorm(0,.001)
  a2 ~ dnorm(0,.001)
  beta ~ dnorm(0,.001)
  for(i in 1:(nind+nz)){
    wt[i] ~ dnorm(mu.wt,tau.wt)
    z[i] ~ dbin(psi,1)
    ff[i] <- z[i]+1
    for(t in 1:T){
      logit(p[i,t]) <- a0*(1-prevcap[i,t]) + a1*prevcap[i,t] + beta*wt[i]
+ a2*lagY[i,t]
      P[i,t,1] <- 0
      P[i,t,2] <- p[i,t]
      mu[i,t] <- P[i,t,ff[i]]
      Y[i,t] ~ dbern(mu[i,t])
    }
  }
  N <- sum(z[1:(nind+nz)])
}

```

Figure 1: *WinBUGS* model specification for the *Microtus* data with behavioral effects and body mass as a covariate on detection probability.

```

model {

  psi~dunif(0,1)
  logmu.gs~dnorm(0,.001)
  lambda<-exp(logmu.gs)
  alpha1~dnorm(0,.001)
  alpha2~dnorm(0,.001)
  beta~dnorm(0,.001)
  for(i in 1:(nind+nzeroes)){
    z[i]~dbin(psi,1)
    gs[i]~dpois(lambda)
    logit(p1[i])<- alpha1 + beta*(1+gs[i])
    logit(p2[i])<- alpha2 + beta*(1+gs[i])
    mu[i,1]<- z[i]*(1-p1[i])*p2[i]
    mu[i,2]<- z[i]*p1[i]*(1-p2[i])
    mu[i,3]<- z[i]*p1[i]*p2[i]
    mu[i,4]<- z[i]*(1-p1[i])*(1-p2[i]) + (1-z[i])
    ncap[i,1:4]~dmulti(mu[i,1:4],1)
  }
  Nc<-sum(z[1:(nind+nzeroes)])
}

```

Figure 2: *WinBUGS* model specification of the double-observer waterfowl survey data where group size is the detection probability covariate.